FAQ 9/16/2011

This FAQ aims to clarify the parameterization and the corresponding Mplus input specification for a Mover-Stayer Latent Transition Model.

1. Introduction

Consider a model where it is hypothesized that stayers have a diagonal transition probability matrix, that is, the probability is 1 that they stay in the same class and the probability is zero that they move to another class. For movers, no restriction on the transition probabilities is hypothesized.

This can be varied in many ways, including the case where one class allows only upward (downward) movement, or where there are no restrictions on either transition matrix, so that the term Mover-Stayer is no longer relevant but a more general mixture LTA is considered. As in UG ex8.14, it is also possible to make stayers have probability one of being in a certain class, and furthermore let this class be the only one that doesn't show movement.

The measurement model can also be varied in many ways. Time-invariant thresholds/means are typically used, with different values for the movers and stayers. UG ex8.14 also lets stayers represent subjects who do no exhibit problem behaviors, whereas movers may exhibit problem behaviors, which for categorical outcomes can be specified in terms of fixed high/low threshold.

Consider an example with two time points and three classes for each time point. Using three classes makes the general case clear. This is specified as

CLASSES = c(2) c1(3) c2(3);

where c is the mover-stayer latent class variable, c1 is the time 1 latent class variable, and c2 is the time 2 latent class variable.

2. Logit parameterization and relationship to transition probabilities

Let Movers correspond to c=1. For c=1 we have the logit formula for the c2 categories x=1, 2, 3 and c1 categories y = 1, 2, 3:

(1) P(c2=x | c1=y, c=1) = [c2#x] + [c2#x ON c1#y] + [c2#x ON c#1] ,

where, as in the page 447 top table of the V6 UG, for the last class of c1, [c2#x ON c1#y]=0 and for the last class of c2, all three terms are zero. The intercepts [c2#x] correspond to "a" in the page 447 table, and the slopes [c2#x ON c1#y] correspond to "b".

Stayers correspond to c=2. For c=2 the logit formula above simplifies because [c2#x ON c#2]=0 due to c being in its last category:

(2) P(c2=x | c1=y, c=2) = [c2#x] + [c2#x on c1#y] .

For Stayers, specifying a diagonal transition matrix requires P(c2=x | c1=y, c=2) to be 1 for x=y and 0 for x different from y. The probabilities of 1 and 0 correspond to +Large and –Large logit values. Note that a zero logit value does not give a zero probability because probabilities are evaluated using the exp function and exp(0)=1. Note also that a negative Large value, can be negated by adding a Very Large positive value as shown below. The 1 and 0 probabilities are accomplished by [c2#x]=-Large and [c2#x on c1#y]=+Large for x=y and –Large for x different from y as follows for the key elements of the c1, c2 table:

Probabilities	Logits
P(c2=1 | c1=1, c=2) =1:	[c2#1@-10] + [c2#1 on c1#1@20] = +10
P(c2=2 | c1=1, c=2) =0:	[c2#2@-10] + [c2#2 on c1#1@-10] = -20
P(c2=1 | c1=2, c=2) =0:	[c2#1@-10] + [c2#1 on c1#2@-10] = -20
P(c2=2 | c1=2, c=2) =1:	[c2#2@-10] + [c2#2 on c1#2@20] = +10
P(c2=1 | c1=3, c=2) =0:	[c2#1@-10] = -10
P(c2=2 | c1=3, c=2) =0:	[c2#1@-10] = -10

Because the last two probabilities are zero, P(c2=3 | c1=3, c=2) =1 due to

P(c2=1 | c1=3, c=2) + P(c2=2 | c1=3, c=2) + P(c2=3 | c1=3, c=2) =1.

All intercept terms are specified in the overall part of the model, which means that Movers inherit the specification [c2#x@-10]. But this does not impose a restriction on the transition probabilities for Movers because of the other two terms in (1) above. Even for the last category of c1, the third term [c2#x ON c#1] is present and freely estimated.

3. Mplus input excerpts

The above example with CLASSES = c(2) c1(3) c2(3); is specified in Mplus as

```
MODEL:
%OVERALL%
c1 c2 ON c;
[c2#1-c2#2@-10];
```

```
MODEL c:
%c#1%   ! Movers
c2 on c1;
%c#2%   ! Stayers
c2#1 ON c1#1@20;
c2#2 ON c1#2@20;
c2#2 ON c1#1@-10;
c2#1 ON c1#2@-10;

MODEL c.c1:
Measurement model specifications, e.g.,
equality across time, but difference across Movers-Stayers

MODEL c.c2:

Measurement model specifications, e.g.,
equality across time, but difference across Movers-Stayers
```

4. A second example: Adding a structure for Movers

Consider the case where there is a hypothesis for the Movers to have equal probability to stay in the same class as to move to any other class. With three classes, each row in the transition table for the Movers then has the probability 0.3333 for each entry. This corresponds to zero logit values. Let the transition probabilities for Stayers have the same structure as in the previous example.

For the Mover class, (1) above gives the logit expression:

P(c2=x | c1=y, c=1) = [c2#x] + [c2#x ON c1#y] + [c2#x ON c#1] .

The sum of these three logit terms is made zero as follows. First of all the terms [c2#x ON c1#y] need to be fixed at zero in the Mover class. The intercept terms [c2#x@-10] are kept as before in the Overall part of the model as was required for Stayers in the previous example. To counter act the -10 intercept terms for the Mover class, the Overall part of the model also needs to include [c2#x ON c#1@10] resulting in the desired zero logits. This works because the c2 ON c terms are zero for Stayers because they refer to the last c class. In this way, the Stayers specification is unchanged and is the same as in the previous example.

The input is therefore:

```
MODEL:
%OVERALL%
c1 on c;
```

```
c2 ON c@10;
[c2#1-c2#2@-10];

MODEL c:
%c#1% ! movers
c2 ON c1@0;
%c#2%  ! stayers
c2#1 ON c1#1@20;
c2#2 ON c1#2@20;
c2#2 on c1#1@-10;
c2#1 on c1#2@-10;
```